

Definiowanie kwadratów, czyli wprowadzenie do programowania artystycznego

Andrzej WALAT



Celem tego artykułu jest pokazanie, że istnieją różne *style programowania* będące wyrazem indywidualnych różnic: upodobań, sposobów atakowania problemów. Nie jest tak, że wszyscy muszą programować według jednego wzorca, bo inaczej zapanowałby chaos, a programy napisane przez jednych byłyby nieczytelne dla innych. Starsi Czytelnicy mogą pamiętać kultową kiedyś publikację Wirtha *Wstęp do programowania systematycznego*. Kiedyś była to ważna książka również dla mnie, ale dziś mam poczucie, że jest to przedmiot kultu trochę już przebrzmiałej i nie mojej religii. Nie chcę powiedzieć, że systematyczność nie jest już ważna. Nadal jest ważna, ale programowanie wymaga jeszcze czegoś, co zbliża je do działalności artystycznej. Nieprzypadkowo inna, bardzo ważna książka **Donalda Knutha** ma tytuł *Sztuka Programowania*.

Elementy postawy artystycznej. Na czym polega istota działań artystycznych? Moim zdaniem ważne są trzy czynniki:

- podejście krytyczne,
- zdolność widzenia problemów z różnych perspektyw,
- stały dialog.

Postawa krytyczna. Wielu ludzi zajmuje się malowaniem, ale tylko niewielu jest wybitnymi artystami. Co decyduje o tym, że człowiek staje się artystą? Zwykle na pierwszym miejscu wymienia się talent i pracowitość, rzadziej krytycyzm. W każdej dziedzinie możemy stawiać sobie małe albo bardzo wysokie wymagania. Możemy zachwycać się byle kreską i plamą, albo nieustannie szukać najdoskonalszych rozwiązań. Znane jest powiedzenie Kamila Pissarro: „Nigdy nie kończę obrazu, dopóki szafa, która na początku była czerwona, nie będzie zielona” – aby uzyskać satysfakcjonujące rozwiązanie, trzeba stale weryfikować pierwsze próby, często radykalnie zmieniać początkowe podejście do problemu. Krytycyzm jest najważniejszym czynnikiem, który odróżnia artystów od grafomanów (nie tylko w sztuce).


Zobaczyć inaczej. Z krytycyzmem wiąże się pytanie: *czy można to zrobić inaczej?* i nieustanne próby spojrzenia na problem z innej perspektywy. Kto obserwował malarzy przy pracy, musiał zauważyć, że często oddalają się od obrazu, żeby obejrzeć go z dystansu świeżym okiem. Często, żeby zobaczyć obraz inaczej, odwracają go do góry nogami. W Muzeum Okręgowym w Bydgoszczy znajduje się oryginalny eksponat „tramwaj Wyczółkowskiego”. Są to sztalugi na szynach. Artysta skonstruował je, kiedy w starszym wieku nie mógł już nieustannie biegać do i od obrazu, zamiast tego kręcił korbką i to obraz odjeżdżał lub się przybliżał. Te przykłady świadczą o tym, jak wiele wysiłku wkładają ludzie twórczy, by oglądać i oceniać wyniki swoich działań z różnych perspektyw.

Dialog. Wybitnych artystów często uważa się za skrajnych indywidualistów realizujących swoje idee w izolacji. Tymczasem historia sztuki mówi coś przeciwnego. Dla artystów kontakt z innymi twórcami i ich dokonaniem miał często większe znaczenie niż kontakt z naturą.

Znane są przykłady nieustannego dialogu wybitnych indywidualności: Van Gogha z Gauguinem, mieszkających i malujących razem w Arles, Braque’a z Picassem, czy wreszcie Picassa z Matissem. Krytycyzm, analizowanie problemów z różnych perspektyw i dialog to ważne cechy postawy artystycznej nie tylko w malarstwie, ale również w innych dyscyplinach nie kojarzonych ze sztuką.

Różne podejścia do rozwiązywania problemów algorytmicznych zademonstruję na przykładzie zadania z zajęć przygotowujących do konkursu informatycznego dla gimnazjalistów pt. LOGIA (Logo i algorytmika) organizowanego corocznie w województwie mazowieckim. Z pozoru jest to elementarne zadanie, w którym trudno odkryć coś szczególnie interesującego. Jednak nawet w tak prostych przypadkach mogą się ujawnić bardzo różne style programowania. Omówię dwa różne rozwiązania Pawła i Gawła, którzy jak w wierszu Fredry mają odmienne osobowości.

Zadanie. Napisz procedurę **osiemkwadratów**, która tworzy na środku ekranu następujący rysunek:



Rys. 1

Rozwiązanie Pawła. Paweł jest uporządkowanym chłopcem. Lubi systematyczne podejście do zadań. Ma poczucie, że nie zrozumie zadania, dopóki nie zobaczy dokładnie jego ogólnej struktury wraz ze wszystkimi szczegółami. Te cechy osobowości widać w jego rozwiązaniu:

```
oto osiemkwadratów
cs czterydwadwakwadratyskok 360
już
```

```
oto czterydwadwakwadratyskok :a
powtórz 4 [dwadwakwadratyskok :a / 3 pw 90]
już
```

```
oto dwadwakwadratyskok :a
pod lw 90 np 3 * :a / 2 pw 90 np :a / 2 opu
dwadwakwadraty :a
pod ws :a / 2 lw 90 ws 3 * :a / 2 pw 90 opu
już
```

```
oto dwadwakwadraty :a
dwakwadraty :a
pod pw 90 np :a lw 90 opu
dwakwadraty :a
pod pw 90 ws :a lw 90 opu
już
```

```
oto dwakwadraty :a
kwadrat :a
pod np :a / 2 pw 45 opu
kwadrat :a / pwk 2
pod pw 45 np :a / 2 opu
zamaluj
pod ws :a / 2 lw 45 opu
pod lw 45 ws :a / 2 opu
już
```

```
oto kwadrat :a
powtórz 4 [np :a pw 90]
już
```

Rozwiązanie Gawła. Gawel podziwia systematyczność Pawła i trochę mu zazdrości. Nie byłby w stanie nawet powtórzyć bez przekłamania takich złożonych słów, jak czterydwadwakwadratyskok. Być może dlatego czuje, że rozumie złożony problem dopiero wtedy, gdy widzi tylko zarys jego ogólnej struktury, bez szczegółów. Jego rozwiązanie jest inne niż Pawła. Oglądając je w oknie pamięci, widzimy tylko dwie procedury.

```
oto osiemkwadratów
pod napoz [-200 -200] opu
kwadrat 400
już
```

```
oto kwadrat :bok
powtórz 4 [naprzód :bok pw 90]
już
```

Wydaje się, że to nie może poprawnie działać, ale po napisaniu polecenia:


```
? osiemkwadratów
```

okazuje się, że działa. Czy nie ma w tym oszustwa? Oszustwa nie, ale trochę prestidigitatorstwa. W oknie pamięci widać tylko ogólną strukturę rozwiązania, a szczegóły są ukryte. Po kliknięciu żółwia prawym przyciskiem myszy i otwarciu okna dialogowego **zmień ż1**, można zobaczyć dwie dodatkowe procedury:

```
oto naprzód :bok
powtórz 2 [oczko :bok / 3]
np :bok / 3
już
```

```
oto oczko :bok
rysuj lista 4 lista :bok 90
np 0.5 * :bok pw 45
wielokąt lista 4 lista 0.7071 * :bok 90
lw 45 np 0.5 * :bok
już
```

Żółw **ż1** jest jednostką klasy **żółw**. W Logomocji każdy obiekt, a w szczególności każdy żółw może mieć zdefiniowane własne procedury określające jego sposób wykonywania różnych zadań. Można nawet przedefiniować procedury pierwotne (jak w tym przypadku **naprzód**). Żółw **ż1** ma własną procedurę wykonywania polecenia **naprzód** (ale polecenie **np** wykonuje w zwykły sposób). Proponuję Czytelnikowi następujący eksperyment. Wyczyść ekran. Zdefiniuj procedury żółwia **ż1**, jak w rozwiązaniu Gawła. Utwórz nowego żółwia **ż2**. Ulokuj **ż1** na pozycji [20 -100], a żółwia **ż2** na pozycji [-320 -100]. Następnie napisz polecenie: **proszę wszystkie** [kwadrat 200]. Oba żółwie wykonają to samo polecenie, ale każdy na swój sposób.



Rys. 2

Definiowanie kwadartów. Rozwiązanie Gawła jest poprawne i może się bardziej podobać niż rozwiązanie Pawła, ale czy możemy zaakceptować użycie przez niego nazwy **kwadrat** dla procedury, która rysuje figury bardziej złożone? Może lepiej użyć innej nazwy, na przykład **kwadart**. Kwadart to kwadrat kreślony przez „żółwia artystę”, który wykonując polecenie **naprzód**, przemieszcza się naprzód o daną liczbę kroków, ale niekoniecznie po odcinku prostoliniowym, tylko na swój własny sposób. Zwykle żółwie, które wykonują **naprzód** w zwykły sposób, zaliczamy do kierunku w sztuce nazywanego minimalizmem. Podobnie jak Kazimierz Malewicz, twórca słynnego czarnego kwadratu na białym tle, kreślą one z wielkim upodobaniem zwykle kwadraty.

Jak to zrobić inaczej? Rozwiązania Pawła i Gawła to tylko dwa spośród setek różnych sposobów rozwiązania tego samego problemu. Na stronie internetowej *Delty* przedstawiam jeszcze rozwiązanie Kasi. W jej rozwiązaniu po wywołaniu odpowiedniej procedury mamy wrażenie, że na ekranie, a ściślej na stronie graficznej *Logomocji*, pojawił się właściwy rysunek, ale faktycznie strona jest czysta, natomiast tworzy się 17 żółwi mających postać kwadratu, które układają się w odpowiedni obraz. Po kliknięciu dowolnego punktu ekranu żółwie rozpoczynają urzekający taniec. Procedura Kasi tworzy ruchome obrazy nazywane **mobilami**.